
Graph Components           
 Graph template 
 Style template 

 Data object 

 Dynamic variables

ods graphics on;
proc reg data=sashelp.class;

model weight = height;
quit;

You can use an SG procedure (SGPLOT, SGPANEL, 
SGSCATTER) to make a graph from raw data or from 
the output data object from a graph. 

You can then use SG Annotation to customize the graph.

(can be modified)

(can be modified 
or in SAS 9.4, overridden) 

(can be output)

Customized Dynamic Variables

SG Annotation: Text and Watermark

SG Annotation: Oval, Line, and Text

7

ODS Document
ods document name=MyDoc (write);
proc reg data=sashelp.class;

ods select fitplot;
model weight=height;

quit;
ods document close;

proc document name=MyDoc;
list / levels=all;

quit;

proc document name=MyDoc;
replay \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1;

quit;

proc document name=MyDoc;
ods exclude dynamics;
ods output dynamics=dynamics;
obdynam \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1;

quit;

Replay from an ODS Document
ods document name=MyDoc (write);
proc reg data=sashelp.class;

ods select fitplot;
model weight=height;

quit;
ods document close;

proc document name=MyDoc;
list / levels=all;

quit;

proc document name=MyDoc;
replay \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1;

quit;

Dynamic Variables
ods document name=MyDoc (write);
proc reg data=sashelp.class;

ods select fitplot;
model weight=height;

quit;
ods document close;

proc document name=MyDoc;
list / levels=all;

quit;

proc document name=MyDoc;
ods exclude dynamics;
ods output dynamics=dynamics;
obdynam \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1;

quit;

proc print; 
where label1 in ( . . . );

run;

Replay with Dynamics Specified
ods document name=MyDoc (write);
proc reg data=sashelp.class;

ods select fitplot;
model weight=height;

quit;
ods document close;

proc document name=MyDoc;
list / levels=all;

quit;

proc document name=MyDoc;
ods exclude dynamics;
ods output dynamics=dynamics;
obdynam \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1;

quit;

proc document name=MyDoc;
replay \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1 /

dynamdata=dynamics;
quit;

11

data dynamics2;
set dynamics;
if label1 = '_SHOWNPARM' 

then nvalue1 = 0;
if label1 = '_SHOWAIC'       

then nvalue1 = 1;
if label1 = '_SHOWSBC' 

then nvalue1 = 1;
if label1 = '_SHOWDEPMEAN' 

then nvalue1 = 1;
if label1 in ('_RSQUARE' 

'_ADJRSQ') 
then nvalue1 = 

round(nvalue1, 0.001);
run;

proc document name=MyDoc;
replay \Reg#1\MODEL1#1\ObswiseStats#1\Weight#1\FitPlot#1 /

dynamdata=dynamics2;
quit;

Dynamics Modified

Naïve Rendering
proc reg data=sashelp.class;

ods output fitplot=fp;
model weight = height;

quit;

proc sgrender data=fp
template=Stat.REG.Graphics.Fit;

run;   

WARNING: The SCATTERPLOT statement will not be drawn because one or more of the 
required arguments were not supplied.

WARNING: The SERIESPLOT statement named 'Fit' will not be drawn because one or 
more of the required arguments were not supplied.

13

Diagnostics Panel Naïve Rendering 

PROC SGRENDER Code with Dynamic Variables 
The next two slides show how to use CALL EXECUTE and 
a DATA step to generate this step:

proc sgrender data=fp template=Stat.REG.Graphics.Fit;

dynamic _SHOWCLM = 1 _SHOWCLI = 1 _WEIGHT = 0 _SHOWSTATS = 1 _NSTATSCOLS = 2 
_SHOWNOBS = 1 _NOBS = 19 _SHOWTOTFREQ = 0 _TOTFREQ = 19 _SHOWNPARM = 1 
_NPARM = 2 _SHOWEDF = 1 _EDF = 17 _SHOWMSE = 1 _MSE = 126.02868962 
_SHOWRSQUARE = 1 _RSQUARE = 0.7705068427 _SHOWADJRSQ = 1 _ADJRSQ = 0.7570072452
_SHOWSSE = 0 _SSE = 2142.4877235 _SHOWDEPMEAN = 0 _DEPMEAN = 100.02631579 
_SHOWCV = 0 _CV = 11.223296526 _SHOWAIC = 0 _AIC = 93.780394884 _SHOWBIC = 0 
_BIC = 96.223301459 _SHOWCP = 0 _CP = 2 _SHOWGMSEP = 0 _GMSEP = 140.9531397 
_SHOWJP = 0 _JP = 139.29486747 _SHOWPC = 0 _PC = 0.2834915472 _SHOWSBC = 0
_SBC = 95.669272843 _SHOWSP = 0 _SP = 7.876793101 _TITLE = "Fit Plot"
_DEPNAME = "Weight" _DEPLABEL = "Weight" _SHORTYLABEL = "Weight"
_SHORTXLABEL = "Height" _CONFLIMITS = "95% Confidence Limits"
_PREDLIMITS = "95% Prediction Limits" _XVAR = "_INDEPVAR1"; 

run;  

15

Primer on CALL EXECUTE
data _null_;

call execute('proc print data=sashelp.class; run;');
run;

data _null_;
input;
call execute(_infile_);
datalines4;

proc 
print 
data
=
sashelp
.
class
; run
;
;;;;

 Use CALL EXECUTE to 
generate PROC SGRENDER 
code to replay the graph and 
use the dynamic variables

 The automatic variable 
_infile_ contains the contents 
of the input buffer

Render with PROC SGRENDER and Dynamics
data _null_;

set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=fp ' ||
'template=Stat.REG.Graphics.Fit;');

call execute('dynamic');
end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, 

quote(trim(cvalue1)))));
if eof then call execute('; run;');

run;

_SHOWCLM = 1
_XVAR = "_INDEPVAR1"

OPTIONS SOURCE;
displays code

17

Annotate the Layout Space
proc template;

delete Stat.REG.Graphics.Fit / store=sasuser.templat;
source Stat.REG.Graphics.Fit / file='temp.tmp';

quit;

data _null_;
infile 'temp.tmp';
input;
if _n_ = 1 then call execute('proc template;');
call execute(_infile_);
if _infile_ =: '   BeginGraph' then bg + 1;
if bg and index(_infile_, ';') then do;

bg = 0;
call execute('annotate;');

end;
run;

BeginGraph;
annotate;

Primer on Drawing Spaces

 Drawing spaces: 'DataPercent', 'DataPixel', 'DataValue', 'GraphPercent', 'GraphPixel', 'LayoutPercent', 
'LayoutPixel', 'WallPercent', and 'WallPixel'

 'LayoutPercent' is the default for PROC SGRENDER ('GraphPercent' for PROC SGPLOT)
 'Graph…' and 'Layout…' work with an ANNOTATE statement in the BEGINGRAPH block
 'Data…' and 'Wall…' produce: WARNING: XSPACE= is invalid. Draw statement discarded

19

Annotate the Layout Space
data anno;

length Label $ 40;
Function = 'Text';     Label  = 'Saturday, July 25, 2015';
Width  = 100;        x1 = 100;   y1 = 104;
Anchor  = 'Right';    TextColor = 'Red'; TextSize = 9;   
output;

Label = 'Confidential - Do Not Distribute';
Width = 150;           x1 = 50;   y1 =  50;   
Anchor = 'Center';
Transparency = 0.8;    TextSize = 28;   Rotate = -41;
output;

run;

proc print;
run;

Default coordinates are percentages of the layout area ('LayoutPercent')

Annotate the Layout Space
data _null_;

set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=fp sganno=anno ' ||
'template=Stat.REG.Graphics.Fit;');

call execute('dynamic');
end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, quote(trim(cvalue1)))));

if eof then call execute('; run;');
run;

proc template;
delete Stat.REG.Graphics.Fit / store=sasuser.templat;

quit;

21

Annotate the Data Space
define statgraph Stat.Reg.Graphics.Fit;

notes "Fit Plot";
dynamic . . .;
BeginGraph;

entrytitle halign=left textattrs=GRAPHVALUETEXT _MODELLABEL halign=center
textattrs=GRAPHTITLETEXT _TITLE " for " _DEPNAME;

layout Overlay / yaxisopts=(label=_DEPLABEL shortlabel=_SHORTYLABEL)
xaxisopts=(shortlabel=_SHORTXLABEL);
. . . 
annotate / id="a";
SCATTERPLOT y=DEPVAR x=_XVAR / markerattrs=GRAPHDATADEFAULT primary=

true rolename=(_tip1=OBSERVATION _id1=ID1 _id2=ID2 _id3=ID3 _id4=
ID4 _id5=ID5) tip=(y x _tip1 _id1 _id2 _id3 _id4 _id5);

SERIESPLOT y=PREDICTEDVALUE x=_XVAR / lineattrs=GRAPHFIT connectorder=
xaxis name="Fit" LegendLabel="Fit";

. . .
endlayout;
. . .

EndGraph;
end;

Annotate the Data Space
data _null_;

infile 'temp.tmp';
input;      
if _n_ = 1                      then call execute('proc template;');
if left(_infile_) =: 'SCATTERPLOT' then call execute('annotate / id="a";');
call execute(_infile_);

run;

23

Annotate the Data Space
data anno(drop=name sex age);

set sashelp.class(where=(name='William') rename=(height=x1 weight=y1));
retain DrawSpace 'DataValue' Function 'Oval' HeightUnit WidthUnit 'Data'

Height 5 Width 1 ID 'a' TextColor LineColor 'Red' TextWeight 'Bold';
output;
function = 'Line';
x1 + 0.5;
y1 + 1;
x2 = x1 + 10;
y2 = y1 + 20;
output;
x1 = x2 - 1;
y1 = y2 + 9;
function = 'Text';
Label = 'William and Mary are the same size!';
Anchor = 'Center';
Width = 6;
output;

run;

SG Annotation Data Set

25

Annotate the Data Space
data _null_;

set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=fp sganno=anno ' ||
'template=Stat.REG.Graphics.Fit;');

call execute('dynamic');
end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, quote(trim(cvalue1)))));

if eof then call execute('; run;');
run;

Multiple Overlays
data _null_;

infile 'temp.tmp';
input;
if _n_ = 1 then call execute('proc template;');
call execute(_infile_);
if index(_infile_, '   layout overlay') then lo + 1;
if lo and index(_infile_, ';') then do;

lo = 0;
lonum + 1;
call execute(catt('annotate / id="LO', lonum, '";'));

end;
run;

define statgraph Stat.Reg.Graphics.DiagnosticsPanel;
. . .
BeginGraph / designheight=defaultDesignWidth;

layout lattice / columns=3 rows=3 ...;
layout overlay / xaxisopts=(shortlabel='Predicted');

annotate / id="LO1";
. . .

endlayout;
layout overlay / xaxisopts=(shortlabel='Predicted');

annotate / id="LO2";
. . .

endlayout;
layout overlay / xaxisopts=(label='Leverage' offsetmax=0.05) 

yaxisopts=(offsetmin=0.05 offsetmax=0.05);
annotate / id="LO3";
. . .

endlayout;
layout overlay / yaxisopts=(label="Residual" 

shortlabel="Resid")
xaxisopts=(label="Quantile");
annotate / id="LO4";
. . .

endlayout;
. . .

endlayout;
. . .

EndGraph;
end;

27

Template Modification and Style Overrides
proc template;

delete Stat.REG.Graphics.Fit / 
store=sasuser.templat;

source Stat.REG.Graphics.Fit / 
file='temp.tmp';

quit;

proc format;
value $sex 'M' = 'Male' 'F' = 'Female';

run;

Template Modification
data _null_;

infile 'temp.tmp';
input;
if _n_ = 1 then call execute('proc template;');
if left(_infile_) =: 'SCATTERPLOT y=DEPVAR' then do;

_infile_ = tranwrd(_infile_, 'markerattrs=GRAPHDATADEFAULT', ' ');
_infile_ = tranwrd(_infile_, '/', '/ group=id1 name="sc"');
end;

if left(_infile_) =: 'BeginGraph' then
_infile_ = 'BeginGraph / attrpriority=none datasymbols=(squarefilled circlefilled);';

call execute(_infile_);
if left(_infile_) =: 'SCATTERPLOT y=DEPVAR' then

call execute('discretelegend "sc" / location=inside across=1 autoalign=(topleft);');
run;

proc reg data=sashelp.class;
ods output fitplot=fp;
model weight=height;
id sex;
format sex $sex.;

quit;

The next slide shows that the 
sex variable is in the data 
object and is named id1

Modify 
scatter 
plot

Write

Add
legend

Style 
override

AttrPriority=None 
enables markers of 
different shapes

29

Data Object

Data Object Modification – What if Sex had not been there? 
data both(drop=height weight rename=(sex=id1));

merge sashelp.class(keep=height weight sex) fp;
if height ne _indepvar1 or weight ne depvar then put _all_;
format sex $sex.;

run;

data _null_;
set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=both');
call execute('template=Stat.REG.Graphics.Fit;');
call execute('dynamic');

end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, quote(trim(cvalue1)))));

if eof then call execute('; run;');
run;

31

PROC GLMSELECT
proc glmselect data=sashelp.baseball plots=coefficients;

class league division;
model logSalary = nAtBat nHits nHome

nRuns nRBI nBB yrMajor|yrMajor
crAtBat|crAtBat crHits|crHits
crHome|crHome crRuns|crRuns
crRbi|crRbi crBB|crBB league 
division nOuts nAssts nError /
selection=forward(stop=AICC 

choose=SBC);
run;

PROC GLMSELECT

33

Create ODS Document and Data Object
ods document name=MyDoc (write);
proc glmselect data=sashelp.baseball plots=coefficients;

ods select CoefficientPanel;
ods output CoefficientPanel=cp;
class league division;
model logSalary = nAtBat nHits nHome nRuns nRBI nBB yrMajor|yrMajor

crAtBat|crAtBat crHits|crHits crHome|crHome crRuns|crRuns
crRbi|crRbi crBB|crBB league division nOuts nAssts nError /
selection=forward(stop=AICC choose=SBC);

run;
ods document close;

proc document name=MyDoc;
list / levels=all;

quit;

Dynamic Variables and Step Chosen
proc document name=MyDoc;

ods exclude dynamics;
ods output dynamics=dynamics;
obdynam \GLMSelect#1\Summary#1\CoefficientPanel#1;

quit;

data _null_;
set dynamics;
if label1 = '_CHOSENVALUE' 

then call symputx('cv', cvalue1);
run;

%put &cv;  

9+CrBB

35

Chosen Parameters and Label of the Last Step 
data chosen(keep=parameter rename=(parameter=parm)); 

set cp(where=(steplabel = "&cv")); 
if standardizedest ne 0;

run;

proc print;
run;

data _null_;
set cp;
call symputx('last', steplabel);

run;

%put &last;   

14+CrHome*CrHome

&cv: 9+CrBB

Modify Data Object
data cp2;

set cp;
rhslabelyvalue = .;
if steplabel = "&last" then do i = 1 to nobs;

set chosen point=i nobs=nobs;
if parm eq parameter then rhslabelyvalue = standardizedest;

end;
run;

proc print; 
where steplabel = "&last";

run;

&last: 14+CrHome*CrHome

37

Modify Template
proc template;

delete Stat.GLMSELECT.Graphics.CoefficientPanel;
source Stat.GLMSELECT.Graphics.CoefficientPanel / file='temp.tmp';

quit;

data _null_;
infile 'temp.tmp';
input;
if _n_ = 1 then call execute('proc template;');
if index(_infile_, 'datalabel=PARAMETER') then

_infile_ = tranwrd(_infile_, 'datalabel',
'markercharacterposition=right markercharacter');

call execute(_infile_);
run;

markercharacterposition=right markercharacter=PARAMETERdatalabel=PARAMETER

Markers           Labels

Make Graph
data _null_;

set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=cp2');
call execute('template=Stat.GLMSELECT.Graphics.CoefficientPanel;');
call execute('dynamic');

end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, 

quote(trim(cvalue1)))));
if eof then call execute('; run;');

run;

39

PROC GLMSELECT with Annotation

Modify the Template
data _null_;

infile 'temp.tmp';
input;
if _n_ = 1 then call execute('proc template;');
if index(_infile_, 'datalabel=PARAMETER') then

_infile_ = tranwrd(_infile_, 'datalabel',
'markercharacterposition=right markercharacter');

call execute(_infile_);
if index(_infile_, ' layout overlay') then lo + 1;
if lo and index(_infile_, ';') then do;

lo = 0;
lonum + 1;
call execute(catt('annotate / id="LO', lonum, '";'));

end;
run;

define statgraph Stat.GLMSelect.Graphics.CoefficientPanel;
. . .
BeginGraph;

layout lattice . . .;
layout overlay . . .;

annotate / id="LO1";
. . . 

endlayout;
if (_SHOWPVAL = 1)

layout overlay . . .;
annotate / id="LO2";
. . . 

endlayout;
else

layout overlay . . .;
annotate / id="LO3";
. . . 

endlayout;
endif;

endlayout;
. . . 

EndGraph;
end;

41

Annotation Data Set
Created by:

 RETAIN statement

 Assignment 
statements

data anno;
length ID $ 3 Function $ 9 Label $ 40;
retain x1Space y1Space x2Space y2Space 'DataPercent' Direction 'In';
length Anchor $ 10 xC1 xC2 $ 20;
retain Scale 1e-12 Width 100 WidthUnit 'Data' CornerRadius 0.8

TextSize 7 TextWeight 'Bold'
LineThickness 1.2 DiscreteOffset -0.3 LineColor 'Green';

ID       = 'LO1';            Function  = 'Text';
Anchor   = 'Right';          TextColor = 'Green';
x1       = 55;               y1        = 94;
Label    = 'Coefficients for the Selected Model';                output;

Function = 'Line';           x1        = .;
x1Space  = 'DataValue';      x2Space   = x1Space;
xc1      = '9+CrBB';         xc2       = '8+CrRuns*CrRuns';
y1       = 94;               y2        = 94;                     output;

Function = 'Rectangle';      y1Space   = 'WallPercent';
Anchor   = 'BottomLeft';     y1        = 10;
Height   = 80;               Width     = 0.6;                    output;

ID       = 'LO3';            Width     = 100;
Function = 'Text ';          Label     = 'Selected Value';
x1Space  = 'DataPercent';    y1Space   = x1Space;
Anchor   = 'Left';           TextColor = 'Blue';
x1       = 86;               y1        = 84;                     output;

Function = 'Arrow';          LineColor = 'Blue';
x1Space  = 'DataValue';      x2Space   = x1Space;
xc1      = '9+CrBB';         xc2       = '12+CrHits*CrHits';
y1       = 4;                y2        = 83;
DiscreteOffset = .1;         x1        = .;                      output;

run;

Annotation Data Set

43

 We processed the dynamic variables but did not modify 
them

 The paper provides macros that make it easy to modify 
graphs

Create Graph
data _null_;

set dynamics(where=(label1 ne '___NOBS___')) end=eof;
if _n_ = 1 then do;

call execute('proc sgrender data=cp2 sganno=anno');
call execute('template=Stat.GLMSELECT.Graphics.CoefficientPanel;');
call execute('dynamic');

end;
if cvalue1 ne ' ' then

call execute(catx(' ', label1, '=',
ifc(n(nvalue1), cvalue1, quote(trim(cvalue1)))));

if eof then call execute('; run;');
run;

Modifications:

 Data object

 Template

 Annotation

 You can modify graph templates

 Use newer appearance options (or modify styles)
 In BEGINGRAPH in GTL or STYLEATTRS in SG procedures

Conclusions

45

Conclusions
 You can capture, display, output, and modify dynamic 

variables; then you can recreate the graph

Conclusions
 You can annotate single- and multiple-panel graphs
 You can do as many or as few modifications as you 

want and in any combination

47

For More Information
 Graphically Speaking SAS Blog

http://blogs.sas.com/content/graphicallyspeaking/

 Advanced ODS Graphics Examples 
http://support.sas.com/documentation/prod-
p/grstat/9.4/en/PDF/odsadvg.pdf

 SAS/STAT introductory chapters  

 Statistical Graphics in SAS
https://support.sas.com/publishing/authors/kuhfeld.html

 Books by Sanjay Matange and Dan Heath
http://support.sas.com/publishing/authors/matange.html

Free!
Free!

Free!

Advanced ODS Graphics Examples
Axes

• Multiple Axes, Offsets, and Drop Lines
• Multiple Axes and Highlighted Points
• Multiple Axes, Axis Alignment, and Many Tick Labels

Axis Tables
• Axis Table Example Using PROC REG
• Creating a Forest Plot Using PROC SGPLOT

Annotation
• Replacing Tick Labels
• Understanding the Drawing Spaces
• Displaying Text in a Graph
• Drawing Lines
• Custom Markers, No Markers, and the Data Region
• Displaying Images in a Graph
• Lines, Circles, Ovals, Rectangles, and Other Shapes
• Watermarks

Bars, Lines, Curves, and Arrows
• Adverse Events Plot
• Attribute Maps

Plots of Labeled Points
• Placing Labels in Scatter Plots

Advanced Customization of Graphs That Analytical Procedures Produce
• Changing Dynamic Variables by Using the ODS Document
• Annotating Single-Panel Graphs That Analytical Procedures Produce

• Broken Axes
• Multiple Plots with Equated Axes

• Stem-and-Leaf Plot with a Box Plot
• Axis Table Example Using PROC AUTOREG

• Rotating Text
• Continuing Text
• Shape and Scale of Arrowheads
• Text Justification and Anchoring
• Selecting the X, X2, Y, and Y2 Axes
• Scaling Images
• Adding Links to Graphs
• SG Annotation Functions, Variables, and Their Values

• Connecting Points with Lines, Arrows, and Curves

• Changing How Vectors Are Displayed

• Annotating Multiple-Panel Graphs That Analytical Procedures Produce

49

